新时代的开端:DELPHI.NET- 语言篇
文/黄忠成(2004/1/1，原文刊登于PC 计算机传真月刊)
DELPHI.NET
在经过近兩年的漫长等待后，Borland 终于在2003 年的耶诞前夕释出了DELPHI 8
for .NET Framework 的正式版本，此举不但粉碎了DELPHI 已死的揣测，也正式将DELPHI
使用者带入了.NET 的世界。藉由VCL.NET，原DELPHI 使用者可以延用以往所开发的
组件或是程序，用几乎相同的方式撰写.NET Windows Application。除此之外，
DELPHI.NET 也支援.NET WinForm 及ASP.NET 的应用程序开发，这就像是DELPHI 一
贯的传统:支持最多的环境，提供多样化的开发技术，让使用者得到最大的应用空间。
当然，不可否认，DELPHI.NET 毕竟是一个新的开发环境及一个加强后的DELPHI 语言，
要說完全无痛升级是不可能的，程序设计师应该要有一个认知，在Windows 时代，API
是不可缺的知識，在今日的.NET 时代也不例外，了解.NET Framework 是基本的要求。
针对DELPHI.NET 所提供的各种新功能及新语言特色，笔者规画了一系列文章，希望
能帮助原DELPHI 的使用者快速的进入DELPHI.NET 的世界，现在就让我们开始这一趟
旅程吧。
Imporved! Namespace(命名空间)
事实上，Namespace 并非是在DELPHI.NET 中第一次出现，早在DELPHI 时代就已具
备基本的Namespace 概念，只是当时此技术尚未成熟，大多數时候只是用來解决模糊呼
叫或是定义等问题，下面的程序代码片段是在DELPHI 时代使用该技术的范例。
Dialogs.ShowMessage(AMsg);
当时，这种技术顶多只能称之为Unit 的分割，離真正的Namespace 尚有段距離。为了
符合CLS 规格，DELPHI.NET 加入了真正的Namespace 机制，使编译后的Assembly 能
为其它CLS 兼容语言所用，其基本用法与以往大致相同。
unit MyCompany.MyProduct.Unit2;
interface
uses Borland.Vcl.Classes,Borland.Vcl.SysUtils;
与DELPHI 相同，DELPHI.NET 的Unit 檔名需与unit ...部份同名，另外原先的Classes、
SysUtils 等Unit 也都纳入了Borland.Vcl 这个Namespace 之中，不过使用者并不需特别
修改以往的程序來符合这个改变，藉由编译器预设Namespace 搜寻功能，使用者仍可使
用uses Classes,SysUtils 方式來处理，编译器自会代为转换。提醒讀者，目前DELPHI.NET
仍只支持在一个Unit 中定义单一Namespace，并不允许同一Unit 拥有多个Namespace。
另外DELPHI.NET Unit 也不支持類似C#等语言中以明确定义代替uses 的方式，简单的
說就是要引用Unit2 才能使用其中的定义，不能省略掉引用Unit2 部份，直接使用
Unit2.xxx 方式宣告。

New! 成员视界
在DELPHI.NET 中，除了原有的private、protected、public、published 四个视界外，还
加入了strict private、strict protected 兩个新的成员视界，此举是为了符合CLS 规格。原先
的DELPHI 中，private、protected 在同一Unit 中是被视同为public 视界的，下面的程序代码
說明了此点。
type
TMyClass = class
private
FData:Integer;
end;
...............
vObj := TMyClass.Create;
vObj.FData := 100; //OK!
这个特性仍被延用至DELPHI.NET，事实上它们被对应至Assembly or Family、Assembly
and Family 兩个CLS 所规范的视界，而strict private 及strict protected 则对应到真实的
private、protected 兩个视界，表一列出目前DELPHI.NET 的视界对应及存取限制。
TMyClass = class
strict private
FPrivateData:Integer;
private
FData:Integer;
end;
.............
vObj := TMyClass.Create;
vObj.FData := 100; //OK
vObj.FPrivateData := 100; //Error
DELPHI.NET C# CLS 說明
private internal Assembly 该類别或同一Assembly 才可存取。
protected internal protected Assembly or Family 同一Assembly 或子類别才可存取。
public public Public 不限。
strict private private Private 该類别中才可存取。
strict protected protected Family 该類别及其子類别才可存取。
New! class var(静态成员变數)
对DELPHI 使用者來說，静态成员变數是个新功能，以往多是以全域变數來达到相同
效果，使用静态成员可以有效的将设计师由全域变數的窠臼中解放。
TMyClass = class(TObject)
class var
InstanceCount:Integer;
..................
constructor TMyClass.Create;
begin
TMyClass.InstanceCount := TMyClass.InstanceCount+1;
inherited Create;
end;
destructor TMyClass.Destroy;
begin
TMyClass.InstanceCount := TMyClass.InstanceCount-1;
inherited Destroy;
end;
..................
TMyClass.Create;
TMyClass.Create;
TMyClass.Create;
ShowMessage(IntToStr(TMyClass.InstanceCount)); // 3.
class var 与一般成员相同，也具备了视界的观念。
New! static member function(静态成员函式)
事实上，DELPHI 中早就具备了静态成员函式的功能，当时称为class method，
DELPHI.NET 除了继续延用此概念外，还加入了static method 的支持，这点主要是针对
CLS 的规格而定。但是目前On-Line Help 在描述语法时似乎有误，其内的范例是在一般
的method 后加上static 即可成为静态成员函式，但实际上编译器并不接受此语法，在笔
者观察VCL 的原始码后，正确的用法应该如下所示。
TMyClass = class(TObject)
class var
InstanceCount:Integer;
public
...............
class procedure ShowInstanceCount; static;
end;
class procedure 是否加上static 的关键词，会影响编译器所产生的IL 程序代码，假如设计
者希望该函式能为其它CLS 兼容语言所用，就必须加上static 关键词。
New! static properties(静态成员属性)
这也是一个全新的功能，DELPHI.NET 允许设计师定义静态的成员属性，其唯一限制
是在属性的存取函式或是变數都必须是静态成员。
TMyClass = class(TObject)
strict private
class procedure SetData(AData:Integer);static;
class function GetData:Integer;static;
class var
FData : Integer;
public
class property Data:Integer read GetData write SetData;
end;
.........
TMyClass.Data := 100;
ShowMessage(IntToStr(TMyClass.Data)); //100
New! static construtcor(静态建构子)
在DELLPHI.NET 中，静态建构子(class constructor)可用來初使化静态成员。
TMyClass = class(TObject)
class var
FDataVar : Integer;
public
class constructor Create;
end;
..............
class constructor TMyClass.Create;
begin
FDataVar := 1000;
end;
......
ShowMessage(IntToStr(TMyClass.FDataVar)); // 1000
基本上，DELPHI.NET 提供了完整的静态成员支持，而其是由原本的class member 所延
伸而來。
New! Nested Type(巢狀型别)
这也是为了符合CLS 所添加的新功能，DELPHI.NET 允许设计师在某个類别中定义
一个子類别，需注意的一点是DELPHI.NET 的Nested Type 并不受视界影响，即使将子
類别宣告于strict private 之中，实际上所编译出來的仍然是public。不过设计师仍可延用
以往将類别宣告于implementation 区段后的方式來达到与private 相同效果。
TMyClass = class(TObject)
public
type
TMyClassInner = class(TObject)
public
FData:Integer;
end;
strict private
FInner : TMyClassInner;
.....................
var
vObj2 : TMyClass.TMyClassInner;
begin
vObj2 := TMyClass.TMyClassInner.Create;
..................
New! Sealed Classes(末代類别)
在CLS 规格中允许设计者定义一个sealed class，此种類别将不支持继承，DELPHI.NET
同时也支持了此功能。不过，DELPHI.NET 的on-line help 对此语法的描述是错误的，
下面是正确的语法。
TMyClass = class sealed
public
procedure HelloMyClass;
end;
TMyClass2 = class(TMyClass) //Error!
New! Operator Overload(运算子覆载)
运算子覆载对于DELPHI 设计师來說是一个完全崭新的概念，藉由此功能的帮助，设
计师可以自定類别的运算子行为，提供使用者一个较直觉的应用模式，例如用”+”符号
直接加总兩个对象实体，或是改变对象的型别转换行为，下面的程序展现了此類运用。
TMyClass = class
strict private
FData:Integer;
public
class operator Add(A,B:TMyClass):TMyClass;
class operator Implicit(A:Integer):TMyClass;
class operator Implicit(A:TMyClass):Integer;
property Data:Integer read FData write FData;
end;
...................
class operator TMyClass.Add(A,B:TMyClass):TMyClass;
begin
Result := TMyClass.Create;
Result.Data := A.Data + B.Data;
end;
class operator TMyClass.Implicit(A:Integer):TMyClass;
begin
Result := TMyClass.Create;
Result.Data := A;
end;
class operator TMyClass.Implicit(A:TMyClass):Integer;
begin
Result := A.Data;
end;
procedure TForm1.FormCreate(Sender: TObject);
var
vObj1,vObj2,vResult,vResult2:TMyClass;
vInt:Integer;
begin
vObj1 := TMyClass.Create;
vObj2 := TMyClass.Create;
vObj1.Data := 100;
vObj2.Data := 100;
vResult := vObj1 + vObj2;
ShowMessage(IntToStr(vResult.Data));
vInt := vResult;
ShowMessage(IntToStr(vInt));
vResult2 := vInt;
ShowMessage(IntToStr(vResult2.Data));
end;
程序中只展示了隐含转型的覆载，DELPHI.NET 也支持明确转型的覆载，只要将Implicit
改为Explicit 即可。DELPHI.NET 的运算子覆载技术完全兼容于CLS 规格，其它支持运
操作数覆载的CLS 语言，如C# 皆可直接引用。不过设计师为此特色兴奋之余，也不能忘
记一点，运算子覆载是个相当严肃的课题，当覆载了”+”符号之后，若不覆载其它运算
符号，将令使用者感到困惑，这也是运算子覆载容易造成错误的原因。
New! class helper
这不仅是一个全新的功能，同时也是一个崭新的概念，DELPHI.NET 为了达到整合
VCL.NET 与.NET Framework 的目的，加入了class helper 功能，运用此功能，设计者可
以在不改变既存類别原始设计的方式，延伸该類别，即使该類别是末代類别也不例外，
下面是一个简单的范例。
TMyClass = class sealed
public
procedure HelloMyClass;
end;
TMyClassHelper = class helper for TMyClass
public
procedure HelloMyClassHelper;
end;
..........
procedure TMyClass.HelloMyClass;
begin
ShowMessage('Hello My Class!');
end;
procedure TMyClassHelper.HelloMyClassHelper;
begin
ShowMessage('Hello My Class Helper!');
end;
...............
var
vObj : TMyClass;
begin
vObj := TMyClass.Create;
vObj.HelloMyClass;
vObj.HelloMyClassHelper;
end;
熟悉Design Patterns 的讀者应该已发现，class helper 有点類似Adapter 及Helper 这兩个
Pattern，是的!基本上class helper 就是这兩个Pattern 的演化，只是DELPHI.NET 采取编
译器角度來实作，并将细节隐藏。class helper 也支持继承与虚拟函式的操作，这代表着
class helper 开启了除继承外的另一扇延伸之窗，下面的范例展示了此种用法。
TMyClassHelper = class helper for TMyClass
public
procedure HelloMyClassHelper;virtual;
end;
TMyClassHelper2 = class helper(TMyClassHelper) for TMyClass
public
procedure HelloMyClassHelper;override;
end;
需特别注意的一点，因为class helper 的实作方式较一般class 不同，因此不支持strict
private、strict protected 这兩个视界，同时也不能拥有成员变數，只允许静态成员变數。
class helper 底层运作
相信有些讀者一定对于class helper 内部的实作方式很好奇，笔者也是，所以利用了
ILDASM 及DeCompiler 工具來观察使用class helper 類别的实作，事实上，class helper
最终的实作码大致与下面的DELPHI 程序代码相同。
TMyClass = class
public
FData:Integer;
end;
TMyClassHelper = class
public
class procedure ShowData(ASelf:TMyClass);
end;
class procedure TMyClassHelper.ShowData(ASelf:TMyClass);
begin
ShowMessage(IntToStr(ASelf.FData));
end;
procedure TForm1.Button1Click(Sender: TObject);
var
vObj : TMyClass;
begin
vObj := TMyClass.Create;
vObj.FData := 100;
TMyClassHelper.ShowData(vObj);
end;
其技术重点在于将class helper 中的函式编译为静态函式，并添加了一个參數，用來传入
helped object。在呼叫端部份的程序代码则会被展开为呼叫该静态函式，这些行为都是由
编译器代勞，下面的程序可以证明这个論点。
procedure TForm1.Button1Click(Sender: TObject);
var
vObj : TMyClass;
vType : System.Type;
Methods : array of MemberInfo;
begin
vObj := TMyClass.Create;
vObj.HelloMyClass;
vObj.HelloMyClassHelper;
{ invoke static function of TMyClassHelper }
vType := TypeOf(TMyClassHelper);
vType.InvokeMember('HelloMyClassHelper',BindingFlags.InvokeMethod,Nil,Nil,[vObj]);
end;
当class helper 含有继承时，其行为会稍微改变，改采Interface 來处理，由于这个课题还
牵扯到了RTL 部份，碍于篇幅，笔者就不再深入讨論了。
PS:InvokeMember 是.NET Reflection 功能的一部份，其功能是由某个型别取出其方法并
呼叫，笔者将其放在下一篇:.NET Framework 中一并讨論。
Next，.NET Framework 篇
本篇文章中，笔者以快速的方式引領讀者浏览DELPHI.NET 所新增的语言特色，不过
由于篇幅有限，实难详述每个特色，例如record 中能定义成员函式等较少用到的功能就
被省略了，讀者们可以參考on-line help 來得到笔者所省略掉的其它细节，下次再見了!
__
